
1 Overview

This document lists all the essential changes needed to correct problems in the TTCN implementation of test case G.12.1 which is part of the iWD-TTCN3-B2016-06_D17wk37 test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 

Contact:
Hans Rohnert

Hans.Rohnert.ext@rohde-schwarz.com
1.1 Verification Test Summary

Test Case:
G.12.1
ATS Version:
iWD-TTCN3-B2016-06_D17wk37
System Simulator used:
Rohde & Schwarz CMW500 Multi-RAT Protocol Tester
UE used:
Samsung SM-G935V-G935VVRE4BQE2
Verification Status:
PASS
2 Corrections required

2.1 f_IMS_InviteResponse_180_MessageHeaderTX
	Function name
	f_IMS_InviteResponse_180_MessageHeaderTX

	Reason for change
	The 180 Ringing message as described in step 4 of C.21a carries an SDP body. Consequently, it needs a Content-Type header, to be provided by the caller, see change regarding f_IMS_MOCallSetup_AnnexC21ac25ab_Steps4_6.

	Summary of change
	Added an input parameter for the Content-Type to use, and populated the Content-Type header accordingly.

	TTCN module
	Common/IMS/IMS_Procedures_CallControl.ttcn

	MCC160 Comment
	


Before change
  function f_IMS_InviteResponse_180_MessageHeaderTX(INVITE_Request p_InviteRequest,

                                                    charstring p_ContactUri := px_IMS_CalleeContactUri,

                                                    boolean p_PrackRequested := true,

                                                    IMS_CallType_Type p_CallType := NORMAL_CALL,

                                                    boolean p_SrvccAlertingSupported := pc_IMS_SRVCCAlert) runs on IMS_PTC return template (value) MessageHeader

  { // p_PrackRequested  ..  condition A3; needs to be true when the 180 response shall be acknowleged with a PRACK i.e. is "sent reliably" acc. to A.2.6 A3

    // p_SrvccAlertingSupported  .. if voice call => condition A6; needs to be set to false for non-voice calls

    /* @sic R5s130333 change 2: new parameter p_PrackRequested sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    /* @sic R5s140370 change 6: new parameter p_SrvccAlertingSupported sic@ */

    /* @sic R5s150692 change 11: parameter p_CallType sic@ */

    var MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (value) MessageHeader v_MessageHeader_Response := cs_MessageHeader_Dummy;

    var template (value) RouteBody_List v_RouteBodyList := f_IMS_RouteSet_MO_Call_TX(p_CallType); /* @sic R5s150692 change 11 sic@ */

    var template (value) SemicolonParam_List v_FeatureParams := {};

    var template (value) SemicolonParam_List v_ContactParams := { cs_GenericParam("audio") };     /* @sic R5-165943, R5-170564, R5s170216 change 1 (R5-173021) sic@ */

    v_MessageHeader_Response.recordRoute     := cs_RecordRoute(v_RouteBodyList);

    v_MessageHeader_Response.via             := v_MessageHeader_Invite.via;

    v_MessageHeader_Response.fromField       := v_MessageHeader_Invite.fromField;

    v_MessageHeader_Response.toField         := f_IMS_Response_ToHeaderTX(v_MessageHeader_Invite); /* @sic R5-131897, R5-132063 - generic handling of to-tags sic@ */

    v_MessageHeader_Response.contact         := cs_Contact(f_SIP_BuildSipUri_TX(p_ContactUri), v_ContactParams); /* @sic R5s130109, R5s170216 change 1 sic@ */

    v_MessageHeader_Response.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Response.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Response.cSeq            := v_MessageHeader_Invite.cSeq;

    if (p_PrackRequested) {     // Condition A3 @sic R5s130333 change 2 sic@

      v_MessageHeader_Response.require       := cs_Require(tsc_OptionTagList_100rel);            /* -> reliable provisional responses acc. RFC 3262 */

      v_MessageHeader_Response.rSeq          := cs_RSeq(tsc_IMS_RSeqNumFor183 + 1);

    }

    select (p_CallType) {

      case (REGULAR_EMERGENCY, ASSERTED_EMERGENCY) { // Condition A4 @sic R5s150692 change 11 sic@

        v_MessageHeader_Response.pAssertedID := cs_PAssertedId(f_SIP_BuildSipUri_lr_TX(tsc_IMS_Emergency_TelUri));

      }

    }

    if (p_SrvccAlertingSupported) {

      v_FeatureParams := f_SemicolonParam_List_Add_TX(v_FeatureParams, cs_GenericParam(tsc_IMS_FeatureCap_SrvccAlerting)); // @sic R5-145796 sic@

    }

    v_MessageHeader_Response.featureCaps   := cs_FeatureCaps(v_FeatureParams);  /* @sic R5s170216 change 1: "audio" removed from FeatureCaps sic@ */

    return v_MessageHeader_Response;

  }
After change
   function f_IMS_InviteResponse_180_MessageHeaderTX(INVITE_Request p_InviteRequest,

                                                    charstring p_ContactUri := px_IMS_CalleeContactUri,

                                                    boolean p_PrackRequested := true,

                                                    IMS_CallType_Type p_CallType := NORMAL_CALL,

                                                    boolean p_SrvccAlertingSupported := pc_IMS_SRVCCAlert,

                                                    template (omit) ContentType p_ContentType := omit)
                                                            runs on IMS_PTC return template (value) MessageHeader

  { // p_PrackRequested  ..  condition A3; needs to be true when the 180 response shall be acknowleged with a PRACK i.e. is "sent reliably" acc. to A.2.6 A3

    // p_SrvccAlertingSupported  .. if voice call => condition A6; needs to be set to false for non-voice calls

    /* @sic R5s130333 change 2: new parameter p_PrackRequested sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    /* @sic R5s140370 change 6: new parameter p_SrvccAlertingSupported sic@ */

    /* @sic R5s150692 change 11: parameter p_CallType sic@ */

    var MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (value) MessageHeader v_MessageHeader_Response := cs_MessageHeader_Dummy;

    var template (value) RouteBody_List v_RouteBodyList := f_IMS_RouteSet_MO_Call_TX(p_CallType); /* @sic R5s150692 change 11 sic@ */

    var template (value) SemicolonParam_List v_FeatureParams := {};

    var template (value) SemicolonParam_List v_ContactParams := { cs_GenericParam("audio") };     /* @sic R5-165943, R5-170564, R5s170216 change 1 (R5-173021) sic@ */

    v_MessageHeader_Response.contentType     := p_ContentType;
    v_MessageHeader_Response.recordRoute     := cs_RecordRoute(v_RouteBodyList);

    v_MessageHeader_Response.via             := v_MessageHeader_Invite.via;

    v_MessageHeader_Response.fromField       := v_MessageHeader_Invite.fromField;

    v_MessageHeader_Response.toField         := f_IMS_Response_ToHeaderTX(v_MessageHeader_Invite); /* @sic R5-131897, R5-132063 - generic handling of to-tags sic@ */

    v_MessageHeader_Response.contact         := cs_Contact(f_SIP_BuildSipUri_TX(p_ContactUri), v_ContactParams); /* @sic R5s130109, R5s170216 change 1 sic@ */

    v_MessageHeader_Response.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Response.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Response.cSeq            := v_MessageHeader_Invite.cSeq;

    if (p_PrackRequested) {     // Condition A3 @sic R5s130333 change 2 sic@

      v_MessageHeader_Response.require       := cs_Require(tsc_OptionTagList_100rel);            /* -> reliable provisional responses acc. RFC 3262 */

      v_MessageHeader_Response.rSeq          := cs_RSeq(tsc_IMS_RSeqNumFor183 + 1);

    }

    select (p_CallType) {

      case (REGULAR_EMERGENCY, ASSERTED_EMERGENCY) { // Condition A4 @sic R5s150692 change 11 sic@

        v_MessageHeader_Response.pAssertedID := cs_PAssertedId(f_SIP_BuildSipUri_lr_TX(tsc_IMS_Emergency_TelUri));

      }

    }

    if (p_SrvccAlertingSupported) {

      v_FeatureParams := f_SemicolonParam_List_Add_TX(v_FeatureParams, cs_GenericParam(tsc_IMS_FeatureCap_SrvccAlerting)); // @sic R5-145796 sic@

    }

    v_MessageHeader_Response.featureCaps   := cs_FeatureCaps(v_FeatureParams);  /* @sic R5s170216 change 1: "audio" removed from FeatureCaps sic@ */

    return v_MessageHeader_Response;

  }
2.2 f_IMS_MOCallSetup_AnnexC21ac25ab_Steps4_6
	Function name
	f_IMS_MOCallSetup_AnnexC21ac25ab_Steps4_6

	Reason for change
	1. As explained in change for f_IMS_InviteResponse_180_MessageHeaderTX, 180 Ringing response needs a Content-Type header with value “application/sdp”. The indication to carry such header is to be fed into the function building the 180 Ringing response.
2. WLAN PTC is waiting for a coordination message

	Summary of change
	1. Added input argument to call to f_IMS_InviteResponse_180_MessageHeaderTX
2. Send co-coordination message after 180 Rining

	TTCN module
	Common/IMS/IMS_Procedures_CallControl.ttcn

	MCC160 Comment
	


Before change
  function f_IMS_MOCallSetup_AnnexC21ac25ab_Steps4_6(IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp,

                                                     charstring p_ContactUri := px_IMS_CalleeContactUri,

                                                     boolean p_PreconditionsRequired := true,

                                                     IMS_SDP_AudioCodec_Type p_AudioCodec := AMR_16000,

                                                     boolean p_SrvccAlertingSupported := pc_IMS_SRVCCAlert,

                                                     IMS_CallType_Type p_CallType := NORMAL_CALL) runs on IMS_PTC

  { /* NOTE: p_AudioCodec and p_PreconditionsRequired are not used for video call */

    var INVITE_Request v_InviteRequest := p_InviteRequestWithSdp.Invite;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(p_InviteRequestWithSdp.RoutingInfo);

    var template (value) MessageHeader v_MessageHeader_180Response := f_IMS_InviteResponse_180_MessageHeaderTX(v_InviteRequest, p_ContactUri, -, p_CallType, p_SrvccAlertingSupported);

    var template (value) MessageHeader v_MessageHeader_PrackResponse_200;

    var template (present) MessageHeader v_MessageHeader_Prack;

    var template (value) SDP_Message v_SDP_MessageStep4;

    var SDP_media_desc v_SDP_Media_AudioRx;

    var SDP_media_desc v_SDP_Media_VideoRx;

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var PRACK_Request v_PrackRequest;

    if (lengthof(p_InviteRequestWithSdp.SdpOffer.media_list) > 1) {     // => video

      v_SDP_Media_AudioRx := p_InviteRequestWithSdp.SdpOffer.media_list[0];

      v_SDP_Media_VideoRx := p_InviteRequestWithSdp.SdpOffer.media_list[1];

      v_SDP_MessageStep4 := f_IMS_BuildSDP_MOCallAudioVideo_Step4(SetupVideoCall, v_SDP_Media_AudioRx, v_SDP_Media_VideoRx);

    }

    else {      // Audio call only

      v_SDP_MessageStep4 := f_IMS_BuildSDP_AnnexC21_Step4(p_InviteRequestWithSdp.SdpOffer, p_PreconditionsRequired, p_AudioCodec);

    }

    // Step 4. Send 180 Ringing

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine180, v_MessageHeader_180Response, cs_MessageBody_SDP(v_SDP_MessageStep4))));

    // Step 5. Receive PRACK

    v_MessageHeader_Prack := f_IMS_PrackRequest_MessageHeaderRX(v_InviteRequest, v_MessageHeader_180Response);

    IMS_Server.receive(car_IMS_Prack_Request(cr_PRACK_Request(f_SIP_BuildSipUri_RX(p_ContactUri), v_MessageHeader_Prack))) -> value v_IMS_DATA_REQ;

    v_PrackRequest := v_IMS_DATA_REQ.Request.Prack;

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    // Step 6. Send 200 OK (response to last PRACK)

    v_MessageHeader_PrackResponse_200 := f_IMS_OtherResponse_200_MessageHeaderTX(v_PrackRequest.msgHeader);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader_PrackResponse_200)));

  }
After change
   function f_IMS_MOCallSetup_AnnexC21ac25ab_Steps4_6(IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp,

                                                     charstring p_ContactUri := px_IMS_CalleeContactUri,

                                                     boolean p_PreconditionsRequired := true,

                                                     IMS_SDP_AudioCodec_Type p_AudioCodec := AMR_16000,

                                                     boolean p_SrvccAlertingSupported := pc_IMS_SRVCCAlert,

                                                     IMS_CallType_Type p_CallType := NORMAL_CALL) runs on IMS_PTC

  { /* NOTE: p_AudioCodec and p_PreconditionsRequired are not used for video call */

    var INVITE_Request v_InviteRequest := p_InviteRequestWithSdp.Invite;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(p_InviteRequestWithSdp.RoutingInfo);

    var template (value) MessageHeader v_MessageHeader_180Response := 

    f_IMS_InviteResponse_180_MessageHeaderTX(v_InviteRequest, p_ContactUri, -, p_CallType, p_SrvccAlertingSupported, cs_ContentTypeSDP); 
    var template (value) MessageHeader v_MessageHeader_PrackResponse_200;

    var template (present) MessageHeader v_MessageHeader_Prack;

    var template (value) SDP_Message v_SDP_MessageStep4;

    var SDP_media_desc v_SDP_Media_AudioRx;

    var SDP_media_desc v_SDP_Media_VideoRx;

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var PRACK_Request v_PrackRequest;

    if (lengthof(p_InviteRequestWithSdp.SdpOffer.media_list) > 1) {     // => video

      v_SDP_Media_AudioRx := p_InviteRequestWithSdp.SdpOffer.media_list[0];

      v_SDP_Media_VideoRx := p_InviteRequestWithSdp.SdpOffer.media_list[1];

      v_SDP_MessageStep4 := f_IMS_BuildSDP_MOCallAudioVideo_Step4(SetupVideoCall, v_SDP_Media_AudioRx, v_SDP_Media_VideoRx);

    }

    else {      // Audio call only

      v_SDP_MessageStep4 := f_IMS_BuildSDP_AnnexC21_Step4(p_InviteRequestWithSdp.SdpOffer, p_PreconditionsRequired, p_AudioCodec);

    }

    // Step 4. Send 180 Ringing

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine180, v_MessageHeader_180Response, cs_MessageBody_SDP(v_SDP_MessageStep4))));

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
    // Step 5. Receive PRACK

    v_MessageHeader_Prack := f_IMS_PrackRequest_MessageHeaderRX(v_InviteRequest, v_MessageHeader_180Response);

    IMS_Server.receive(car_IMS_Prack_Request(cr_PRACK_Request(f_SIP_BuildSipUri_RX(p_ContactUri), v_MessageHeader_Prack))) -> value v_IMS_DATA_REQ;

    v_PrackRequest := v_IMS_DATA_REQ.Request.Prack;

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    // Step 6. Send 200 OK (response to last PRACK)

    v_MessageHeader_PrackResponse_200 := f_IMS_OtherResponse_200_MessageHeaderTX(v_PrackRequest.msgHeader);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader_PrackResponse_200)));

  } 

2.3 f_IMS_BuildSDP_AnnexC21_Step4
	Function name
	f_IMS_BuildSDP_AnnexC21_Step4

	Reason for change
	According to TS 34.229-1, C.21a, step 4, the direction-tag of the current-status for the remote leg is “sendrecv” rather than “none”

	Summary of change
	Replaced “none” by “sendrecv”

	TTCN module
	Common/IMS/IMS_SDP_Messages.ttcn

	MCC160 Comment
	


Before change
  function f_IMS_BuildSDP_AnnexC21_Step4(SDP_Message p_SDP_MessageInvite,

                                         boolean p_PreconditionsRequired := true,

                                         IMS_SDP_AudioCodec_Type p_AudioCodec := AMR_16000,

                                         integer p_BandwidthAS := 37)

    runs on IMS_PTC

    return template (value) SDP_Message

  { /* @sic R5-135020: parameter p_SDP_MessageInvite to get RR and RS values sic@ */

    /* @sic R5-144402: p_IPAddrOrigin removed sic@ */

    /* @sic R5s141244: use of f_IMS_BuildSDP_TX sic@ */

    /* @sic R5-151968: p_IsC44 sic@ */

    /* @sic R5s150704, R5s150751 - Additional changes: p_PreconditionsRequired sic@ */

    /* @sic R5-153999: p_IsC44 renamed to p_AudioCodec sic@ */

    /* @sic R5-168031: p_AudioCodec - default value changed from AMR_8000 to AMR_16000 sic@ */

    /* @sic R5-168031: p_BandwidthAS sic@ */

    var template (omit) SDP_attribute v_Attribute;

    var SDP_media_desc v_SDP_Media_Invite := p_SDP_MessageInvite.media_list[0];

    var template (present) SDP_attribute_fmtp_format_specific_params v_FmtpParameters := f_SDP_FmtpDefaultParametersAudioCodecRX(p_AudioCodec);   /* @sic R5s160379 change 2 sic@ */

    var charstring v_FmtAudio := f_SDP_MediaDescr_GetAudioCodec(v_SDP_Media_Invite, p_AudioCodec, v_FmtpParameters);

    var integer v_BandwidthRS := f_SDP_BandwidthList_GetBandwidth(v_SDP_Media_Invite.bandwidth, "RS");

    var integer v_BandwidthRR := f_SDP_BandwidthList_GetBandwidth(v_SDP_Media_Invite.bandwidth, "RR");

    var template (value) SDP_bandwidth_list v_Bandwidth_List_Media := cs_SDP_Bandwidth_List_Media(p_BandwidthAS, v_BandwidthRS, v_BandwidthRR); // @sic R5-123505, R5-135020; R5-144700: AS=p_BandwidthAS sic@

    var template (omit) SDP_attribute_list v_SDP_MediaAttributes;

    var template (omit) SDP_attribute_list v_SDP_PrecondionAttributes := omit;

    var template (omit) SemicolonParam_List v_FmtpParameterList_Step4 := omit;

    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();  /* @sic R5-171477 sic@ */

    var SemicolonParam_List v_FmtpParameterList_Step2;

    var CharStringList_Type v_FmtpParametersToBoCopied;

    var charstring v_ReceivedBw;

    select (p_AudioCodec) {    // @sic R5-151968, R5-153999 sic@

      case (EVS_16000){     // C.44 @sic R5-153999 sic@

        v_Attribute := f_SDP_AttributeList_GetAttribute(v_SDP_Media_Invite.attributes, cr_SDP_Attribute_fmtp(v_FmtAudio));

        v_ReceivedBw := f_SIP_SemicolonParamList_GetParamValue(v_Attribute.fmtp.params.paramList, "bw");   /* @sic R5s170336/R5-172069 sic@ */

        v_FmtpParameterList_Step4 := {cs_EVS_Bw(v_ReceivedBw), cs_EVS_BwSend(v_ReceivedBw), cs_EVS_BwRecv(v_ReceivedBw), cs_AMR_MaxRed};   /* @sic R5s160379: Additional Change by MCC160 sic@

                                                                                                                                              @sic R5s170336/R5-172069: v_ReceivedBw sic@ */

        v_FmtpParametersToBoCopied := { "br", "br-send", "br-recv" };

        v_FmtpParameterList_Step2 := f_SDP_MediaDescr_GetCodecParameters(v_SDP_Media_Invite.attributes, v_FmtAudio);

        v_FmtpParameterList_Step4 := f_SIP_SemicolonParamList_CopyParams(v_FmtpParameterList_Step2, v_FmtpParameterList_Step4, v_FmtpParametersToBoCopied);

        v_SDP_MediaAttributes := f_SDP_MediaAttributes_EVS_AudioDef(v_FmtAudio, valueof(v_FmtpParameterList_Step4));   // valueof to avoid compiler warning

      }

      case (AMR_16000){       // @sic R5-153950 sic@

        v_SDP_MediaAttributes := f_SDP_MediaAttributes_AMR_WB_AudioDef(v_FmtAudio);  /* @sic R5s160379 change 2 sic@ */

      }

      case (AMR_8000){

        v_SDP_MediaAttributes := f_SDP_MediaAttributes_AudioDef(v_FmtAudio);   /* @sic R5-144693: new function f_SDP_MediaAttributes_AudioDef sic@ */

      }

    }

    if (f_SDP_AttributeList_CheckAttribute(v_SDP_Media_Invite.attributes, cr_SDP_Attribute_inactive)) {               /* @sic R5-150694 sic@ */

      v_SDP_MediaAttributes := f_SDP_Attributes_Add_TX(v_SDP_MediaAttributes, cs_SDP_Attribute_inactive);

    }

    if (not v_IsWLAN) { // @sic R5-171476: not required for C.21a sic@

      v_SDP_MediaAttributes := f_IMS_SDP_MediaAttributes_CopyEcnCapabilityAttributesTX(v_SDP_Media_Invite, v_SDP_MediaAttributes);

    }

    v_SDP_MediaAttributes := f_IMS_SDP_MediaAttributes_AddMediaSecurityAttributesTX(v_SDP_MediaAttributes);  /* @sic R5-160293,  R5-160915 new TC H.12.3 sic@ */

    if (p_PreconditionsRequired) {  /* @sic R5s150704, R5s150751 - Additional changes sic@ */

      if (v_IsWLAN) {               /* @sic R5-171477 - WLAN Annex C.21a sic@ */

        v_SDP_PrecondionAttributes := cs_SDP_PrecondionAttributes(c_sendrecv, c_none, c_mandatory, c_mandatory);

      } else {

        v_SDP_PrecondionAttributes := cs_SDP_PrecondionAttributes(c_none, c_none, c_mandatory, c_mandatory);

        v_SDP_PrecondionAttributes := f_SDP_Attributes_Add_TX(v_SDP_PrecondionAttributes, cs_SDP_Attribute_conf_qos);   // @sic R5s120530 change 3.5 sic@

      }

    }

    return f_IMS_BuildSDP_TX(cs_SDP_Media_Audio(v_FmtAudio), v_Bandwidth_List_Media, v_SDP_MediaAttributes, v_SDP_PrecondionAttributes);

  }  
After change
  function f_IMS_BuildSDP_AnnexC21_Step4(SDP_Message p_SDP_MessageInvite,

                                         boolean p_PreconditionsRequired := true,

                                         IMS_SDP_AudioCodec_Type p_AudioCodec := AMR_16000,

                                         integer p_BandwidthAS := 37)

    runs on IMS_PTC

    return template (value) SDP_Message

  { /* @sic R5-135020: parameter p_SDP_MessageInvite to get RR and RS values sic@ */

    /* @sic R5-144402: p_IPAddrOrigin removed sic@ */

    /* @sic R5s141244: use of f_IMS_BuildSDP_TX sic@ */

    /* @sic R5-151968: p_IsC44 sic@ */

    /* @sic R5s150704, R5s150751 - Additional changes: p_PreconditionsRequired sic@ */

    /* @sic R5-153999: p_IsC44 renamed to p_AudioCodec sic@ */

    /* @sic R5-168031: p_AudioCodec - default value changed from AMR_8000 to AMR_16000 sic@ */

    /* @sic R5-168031: p_BandwidthAS sic@ */

    var template (omit) SDP_attribute v_Attribute;

    var SDP_media_desc v_SDP_Media_Invite := p_SDP_MessageInvite.media_list[0];

    var template (present) SDP_attribute_fmtp_format_specific_params v_FmtpParameters := f_SDP_FmtpDefaultParametersAudioCodecRX(p_AudioCodec);   /* @sic R5s160379 change 2 sic@ */

    var charstring v_FmtAudio := f_SDP_MediaDescr_GetAudioCodec(v_SDP_Media_Invite, p_AudioCodec, v_FmtpParameters);

    var integer v_BandwidthRS := f_SDP_BandwidthList_GetBandwidth(v_SDP_Media_Invite.bandwidth, "RS");

    var integer v_BandwidthRR := f_SDP_BandwidthList_GetBandwidth(v_SDP_Media_Invite.bandwidth, "RR");

    var template (value) SDP_bandwidth_list v_Bandwidth_List_Media := cs_SDP_Bandwidth_List_Media(p_BandwidthAS, v_BandwidthRS, v_BandwidthRR); // @sic R5-123505, R5-135020; R5-144700: AS=p_BandwidthAS sic@

    var template (omit) SDP_attribute_list v_SDP_MediaAttributes;

    var template (omit) SDP_attribute_list v_SDP_PrecondionAttributes := omit;

    var template (omit) SemicolonParam_List v_FmtpParameterList_Step4 := omit;

    var boolean v_IsWLAN := f_IMS_PTC_RanTypeIsWLAN();  /* @sic R5-171477 sic@ */

    var SemicolonParam_List v_FmtpParameterList_Step2;

    var CharStringList_Type v_FmtpParametersToBoCopied;

    var charstring v_ReceivedBw;

    select (p_AudioCodec) {    // @sic R5-151968, R5-153999 sic@

      case (EVS_16000){     // C.44 @sic R5-153999 sic@

        v_Attribute := f_SDP_AttributeList_GetAttribute(v_SDP_Media_Invite.attributes, cr_SDP_Attribute_fmtp(v_FmtAudio));

        v_ReceivedBw := f_SIP_SemicolonParamList_GetParamValue(v_Attribute.fmtp.params.paramList, "bw");   /* @sic R5s170336/R5-172069 sic@ */

        v_FmtpParameterList_Step4 := {cs_EVS_Bw(v_ReceivedBw), cs_EVS_BwSend(v_ReceivedBw), cs_EVS_BwRecv(v_ReceivedBw), cs_AMR_MaxRed};   /* @sic R5s160379: Additional Change by MCC160 sic@

                                                                                                                                              @sic R5s170336/R5-172069: v_ReceivedBw sic@ */

        v_FmtpParametersToBoCopied := { "br", "br-send", "br-recv" };

        v_FmtpParameterList_Step2 := f_SDP_MediaDescr_GetCodecParameters(v_SDP_Media_Invite.attributes, v_FmtAudio);

        v_FmtpParameterList_Step4 := f_SIP_SemicolonParamList_CopyParams(v_FmtpParameterList_Step2, v_FmtpParameterList_Step4, v_FmtpParametersToBoCopied);

        v_SDP_MediaAttributes := f_SDP_MediaAttributes_EVS_AudioDef(v_FmtAudio, valueof(v_FmtpParameterList_Step4));   // valueof to avoid compiler warning

      }

      case (AMR_16000){       // @sic R5-153950 sic@

        v_SDP_MediaAttributes := f_SDP_MediaAttributes_AMR_WB_AudioDef(v_FmtAudio);  /* @sic R5s160379 change 2 sic@ */

      }

      case (AMR_8000){

        v_SDP_MediaAttributes := f_SDP_MediaAttributes_AudioDef(v_FmtAudio);   /* @sic R5-144693: new function f_SDP_MediaAttributes_AudioDef sic@ */

      }-
    }

    if (f_SDP_AttributeList_CheckAttribute(v_SDP_Media_Invite.attributes, cr_SDP_Attribute_inactive)) {               /* @sic R5-150694 sic@ */

      v_SDP_MediaAttributes := f_SDP_Attributes_Add_TX(v_SDP_MediaAttributes, cs_SDP_Attribute_inactive);

    }

    if (not v_IsWLAN) { // @sic R5-171476: not required for C.21a sic@

      v_SDP_MediaAttributes := f_IMS_SDP_MediaAttributes_CopyEcnCapabilityAttributesTX(v_SDP_Media_Invite, v_SDP_MediaAttributes);

    }

    v_SDP_MediaAttributes := f_IMS_SDP_MediaAttributes_AddMediaSecurityAttributesTX(v_SDP_MediaAttributes);  /* @sic R5-160293,  R5-160915 new TC H.12.3 sic@ */

    if (p_PreconditionsRequired) {  /* @sic R5s150704, R5s150751 - Additional changes sic@ */

      if (v_IsWLAN) {               /* @sic R5-171477 - WLAN Annex C.21a sic@ */

        v_SDP_PrecondionAttributes := cs_SDP_PrecondionAttributes(c_sendrecv, c_sendrecv, c_mandatory, c_mandatory);
      } else {

        v_SDP_PrecondionAttributes := cs_SDP_PrecondionAttributes(c_none, c_none, c_mandatory, c_mandatory);

        v_SDP_PrecondionAttributes := f_SDP_Attributes_Add_TX(v_SDP_PrecondionAttributes, cs_SDP_Attribute_conf_qos);   // @sic R5s120530 change 3.5 sic@

      }

    }

    return f_IMS_BuildSDP_TX(cs_SDP_Media_Audio(v_FmtAudio), v_Bandwidth_List_Media, v_SDP_MediaAttributes, v_SDP_PrecondionAttributes);

  }  

2.4 f_WLAN_TurnCellOn
	Function name
	f_WLAN_TurnCellOn

	Reason for change
	When BeaconRSSI value is taken from px_WLAN_BeaconRSSI it should not be subjected to the conversion function anymore.

	Summary of change
	Only subject the default value of -50 dbm to the conversion.

	TTCN module
	Common/WLAN/WLAN_CommonProcedures.ttcn

	MCC160 Comment
	


Before change
  function f_WLAN_TurnCellOn(WLAN_CellId_Type p_CellId,

                             boolean p_IMSoWLAN := false) runs on WLAN_PTC

  {

    var integer v_PowerLevel := -50;
    if(p_IMSoWLAN){

      v_PowerLevel:= px_WLAN_BeaconRSSI; // @sic R5-174499  sic@

    }

    f_WLAN_SetCellPower(p_CellId, f_WLAN_ConvertdBm2BeaconRSSI(v_PowerLevel));

  }

After change
  function f_WLAN_TurnCellOn(WLAN_CellId_Type p_CellId,

                             boolean p_IMSoWLAN := false) runs on WLAN_PTC

  {

    var WLAN_BeaconRSSI_Type v_PowerLevel := f_WLAN_ConvertdBm2BeaconRSSI(-50);
    if(p_IMSoWLAN){

      v_PowerLevel:= px_WLAN_BeaconRSSI; // @sic R5-174499  sic@

    }

    f_WLAN_SetCellPower(p_CellId, v_PowerLevel);

  } 
2.5 a_WLAN_Indication
	Function name
	a_WLAN_Indication

	Reason for change
	We need a new alt step in order to handle the SS indication at switch off.

	Summary of change
	New altstep has been created in order to use it as a temporary default.

	TTCN module
	WLAN_CommonProcedures.ttcn

	MCC160 Comment
	


Before change
  n/a
After change
  altstep a_WLAN_Indication (WLAN_CellId_Type p_CellId, timer p_Watchdog) runs on WLAN_PTC

  {

      [] SYS_CMDIND.receive(car_WLAN_SYSTEM_IND(p_CellId, cr_WLAN_AP_DisassociationCmplInd)) {

      }

      [] SYS_CMDIND.receive(car_WLAN_SYSTEM_IND(p_CellId, cr_WLAN_ErrorInd)) {

        f_WLAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "error while waiting for indication from WLAN AP");

      }

      [] p_Watchdog.timeout {

        f_WLAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "timeout while waiting for indication from WLAN AP");

      }

  }
2.6 f_WLAN_Postamble
	Function name
	f_WLAN_Postamble

	Reason for change
	WLAN_AP_DissassociateCmpl indication should not be mandatory at switch off.

According to 36.523-3 annex D.13.3 “WLAN_ePDG_RoutingInfo_Type” should be used as “None” when it is not specific to a particular ePDG PDN Request or Configuration.  

	Summary of change
	1. Created new default variable in order to handle a temporary default to optionally received SS Indication.

2. Removed a specific PDN in WLAN Cell Release.

	TTCN module
	Common/WLAN/WLAN_CommonProcedures.ttcn

	MCC160 Comment
	


Before change
  function f_WLAN_Postamble(WLAN_CellId_Type p_CellId,

                            PDN_Index_Type p_PDN_Index,

                            boolean p_IMSoWLAN := false) runs on WLAN_PTC

  {

    timer t_Watchdog := tsc_WLANWatchDogDef;

    if (pc_SwitchOnOff) {

      f_UT_SwitchOffUE (UT, false);

      t_Watchdog.start;

      //IpSec Tunnel release acc. 36.523-3 cl. 10.6.3.1

      f_WLAN_IPsecTunnel_WaitForIndication(p_PDN_Index, t_Watchdog, cr_WLAN_IPsecTunnelDiscInd);

      //IpSec Tunnel release acc. 36.523-3 cl. 10.6.3.2

      f_WLAN_AP_WaitForIndication(t_Watchdog, p_CellId, cr_WLAN_AP_DisassociationCmplInd);
      t_Watchdog.stop;

    } else {

      f_UT_PowerOffUE (UT);

    }

    f_WLAN_ReleaseCells(p_PDN_Index, p_CellId, -, -, p_IMSoWLAN);

  } 
After change
 function f_WLAN_Postamble(WLAN_CellId_Type p_CellId,

                            PDN_Index_Type p_PDN_Index,

                            boolean p_IMSoWLAN := false) runs on WLAN_PTC

  {

    timer t_Watchdog := tsc_WLANWatchDogDef;

    var default v_Wlan_Dis;
    if (pc_SwitchOnOff) {

      f_UT_SwitchOffUE (UT, false);

      t_Watchdog.start;

      //IpSec Tunnel release acc. 36.523-3 cl. 10.6.3.1

      f_WLAN_IPsecTunnel_WaitForIndication(p_PDN_Index, t_Watchdog, cr_WLAN_IPsecTunnelDiscInd);

      //IpSec Tunnel release acc. 36.523-3 cl. 10.6.3.2      
      v_Wlan_Dis := activate(a_WLAN_Indication(p_CellId,t_Watchdog));
      t_Watchdog.stop;

    } else {

      f_UT_PowerOffUE (UT);

    }

    f_WLAN_ReleaseCells(-, p_CellId, -, -, p_IMSoWLAN);

    deactivate (v_Wlan_Dis);
  }
3 Branches executed
n/a
4 Execution Log Files

4.1 Samsung SM-G935V-G935VVRE4BQE2
The Samsung SM-G935V-G935VVRE4BQE2 UE passed this test case on Rohde & Schwarz CMW500 Multi-RAT Protocol Tester. The documentation below is enclosed as evidence of the successful test case run [1]

